A general pattern: fold
(Adapted from CSE 505 lecture notes by Craig Chambers.)
Many recursive functions on lists follow a common pattern:
fun rev nil = nil
  | rev (x::xs) = (rev xs)::x;
-----
To recurse on an 'a list, you specify:
· what to do as the base case result ('b)
· how to compute the inductive result from the head and the recursive call  ('a * 'b -> 'b)
-----
fold captures this pattern:
fun foldl f base lst =
  case lst of
      [] => base
    | hd::tl => foldl f (f (hd,base)) tl;
-----
Q:  What is the type of foldl?
A:  ('a * 'b -> 'b) -> 'b -> 'a list -> 'b
Q:  What about the definition of foldr?
A:
fun foldr f base lst =
  case lst of
      [] => base
    | hd::tl => f (hd,(foldr f base tl));
-----
Q:  What is the type of foldr?
A:  ('a * 'b -> 'b) -> 'b -> 'a list -> 'b
-----
Q:  What do these do?
- foldl (fn (x,ls) => x::ls) nil [3,4,5];
- foldr (fn (x,ls) => x::ls) nil [3,4,5];
- foldr (fn (x,ls) => x::ls) [1,2,3] [4,5,6];
- foldl (fn (x,ls) => ls@[x]) [1,2,3] [4,5,6];
A:
[5,4,3]
[3,4,5]
[4,5,6,1,2,3]
[1,2,3,4,5,6]
So, using fold, we can define list reverse:
- fun rev lst = foldl (op ::) nil lst;
val rev = fn : 'a list -> 'a list
And list append:
- fun append (l1,l2) =
=   foldl (fn (x,ls) => ls@[x]) l1 l2;
val append = fn : 'a list * 'a list -> 'a list
-----
Q:  How would we define map using fold?  What is map's type?
- fun map f lst = ...
A:
- fun map f lst =
=   foldr (fn (x,ls) => (f x)::ls) nil lst;
val map = fn : ('a -> 'b) -> 'a list -> 'b list
or, equivalently,
- fun map f =
=   foldr (fn (x,ls) => (f x)::ls) nil;
val map = fn : ('a -> 'b) -> 'a list -> 'b list
